

PPC 與 XML Web Service 的整合應用

作者：吳靜如 Lulu Wu

(lulu@soon-get.com)

發表日期：2001 年 12 月

相信您對 PocketPC 這個名稱並不會感到陌生，它漸漸取代記事本、日記本、記帳簿...等等，不但比筆記型電腦更容易攜帶，而且不需要等待開機、關機的時間，越來越符合現代人的需要，隨時隨地，可以管理自己的任何瑣事及時間，就像是個人秘書一般。

在前一期裡，我們已經介紹過開發 XML Web Service 與 Outlook 的應用，現在我們把已經建好了 XML Web Service，讓它達到最大的效益，展現出 XML Web Service 跨平台的魅力。讓您的服務，不再受限於載具的不同。

以 XML Web Service 整合 Pocket Outlook 行事曆之實例

功能說明：

利用 Pocket PC 中的行事曆，來幫助我們整合各種不同資源的最新訊息如：研討會，各式說明會、證券之股東說明會、證券除息除權及台灣微軟各項公開活動，透過 XML Web Service 將各種資訊整合入 Pocket PC 中的行事曆，避免浪費過多的時間與資源在廣大網路海中找尋所需的資訊。

範例下載：

<http://www.msdev.com.tw/search/DownloadDirSearch.aspx?dir=./dotNet/PocketSchedule>

操作說明：

選擇您欲查詢的[期間]及提供服務的[提供者]之後，會將提供者這期間之內的所有活動項目列在下方，並列出各活動項目的主題、時間、地點、提供者。若有想要參加的活動，只要選擇活動項目，再點選[新增至行事曆](見圖 1)，會出現確認的訊息視窗，按下[是](見圖 2)，出現已成功新增的訊息視窗(見圖 3)，便可以在您的行事曆中看到這項活動(見圖 4)。



圖 1 · 列出所有的活動




圖 2 · 確認新增的訊息視窗




圖 3 · 新增成功的訊息視窗



圖 4 · 成功的新增至行事曆中

實作說明

在這個實例中，於 Form_onload 的事件中，我們先建立選單列表(MenuBar)並將提供者的名稱及網址以陣列的方式存至全域變數 strProviders 中，在變更提供者或期間或使用者點選選單[重新連線]時，呼叫相對於提供者的 XML Web Service，將其回傳資料解析後，列表於 TreeView 控制項中，當使用者選擇某一項活動並點選選單[新增至行事曆]時(見圖 1)，我們利用 Pocket Outlook Object Model 新增這項約會至行事曆中。接下來，我將每一個功能獨立成一個函式，方便程式中的互相溝通。

在開始之前，您必須要有以下其中一項工具。

- Microsoft eMbedded Visual Basic,Version 3.0
(可至微軟網站上免費下載 <http://www.microsoft.com/mobile/downloads/emvt30.asp>)
- Microsoft Windows CE Toolkit for Visual Basic 6.0,Version 1.0

PocketPC 如何呼叫 XML Web Service ?

在 PC 上，您可以使用任何的語言利用 SOAP Client 物件去呼叫 XML Web Service。在 PocketPC 上，您可以利用 XMLHttp 物件或是協力廠商(third-party)的產品(PocketSOAP、ViaXML...)來呼叫 XML Web Service。由於版面受限，所以只介紹如何使用 XMLHttp 物件呼叫 XML Web Service。

關於更多的 PocketSOAP 資訊：<http://www.pocketsoap.com/pocketsoap/>

關於更多的 viaXML 資料：<http://www.odysseysoftware.com/>

使用 XMLHttp 物件呼叫 XML Web Service

利用 XMLHttp 物件呼叫 XML Web Services 後，將其回傳的 XML 字串，再傳入 XML DOM 元件並將所需的資料解析出來。目前 Windows CE 3.0 上只支援到 Microsoft XML 2.0，而且與 PC 版的 XML 2.0 有小部份的差異。

加入 XML 參考

Windows CE 上支援的 Microsoft XML2.0 與 PC 版的只有少部份的不同，為了使 MSXML 的智慧感應(IntelliSense)功能啟動，您可以在 eVB 中加入 PC 版的 MSXML2.0 dll。

由於要加入的參考是屬於 PC 上的 MSXML 元件，所以您將不會在有效的參考(Available Reference)下找到[Microsoft XML,version 2.0] 程式庫，如果您有安裝 MSXML2.0 的話，您可以在您的 C:\winnt\system32\目錄下找到 msxml.dll 檔案。按下[Project]→[References] → [Browse...]，選擇 c:\winnt\system32\msxml.dll，按下[開啓]。(見圖 5)




圖 5 · 選擇\winnt\system32\msxml.dll，並按下[開啓]

由於這個不是屬於 windows CE 平台的 dll，所以會出現一個確認視窗，請忽略它並按下[是]。(見

圖 6)




圖 6 · 元件因平台不同而出現的確認視窗，請按下[是]

您將可以把 Microsoft XML,Version 2.0 程式庫參考進來，找到程式庫後，將名稱左方的核取方塊打勾，再按下[OK]鍵離開此畫面。(見圖 7)




圖 7 · 將 XML 加入參照

呼叫 XML Web Service

利用 XMLHttpRequest 物件去呼叫 XML Web Service 的網頁，若網頁執行正確，則解析 XMLHttpRequest 物件回傳的資料，並將每一個活動內容新增至 TreeView 中。在這裡由於程式需要重覆呼叫，所以將其分成 GetWebService、ParseAppItems 這二個函式，由 GetWebService 函式(程式碼 1)取得 XML Web Service 的回傳資料，並傳至 ParseAppItems 函式(程式碼 2)來將資料解析為各項活動再依序新增至 TreeView 中。

```
Private Sub GetWebService(ByVal strURL As String)
    '傳入 Webservice 的網址，例如： http://61.222.11.234/MSTCWebService.asmx?wsdl
    Dim objXML As XMLHttpRequest
    Dim strTemp As String
```

```

Set objXML = CreateObject("Microsoft.xmlHTTP")
objXML.open "GET", strURL, False
objXML.send
If objXML.Status = 200 Then
    '假如 XML 回傳的狀態為 200，代表網頁執行正常
    '回傳的資料本身就為 XMLDOM，所以可以依其的樹狀結構取得資料
    strTemp = objXML.responseXML.childNodes.Item(1).childNodes.Item(0).nodeValue
    Call ParseAppItems(strTemp) '見 程式碼 2
Else
    '假如 XML 回傳的狀態不為 200，代表網頁執行發生錯誤
    txtStatus.Text = "連線失敗..."
    txtStatus.Visible = True
End If
End Sub

```

程式碼 1 · GetWebService 函式

```

Private Function ParseAppItems(ByVal strXML As String) As Boolean
    Dim nTemp As Integer, nCount As Integer, strTemp As String
    Dim strSubject As String, strProvider As String, strLocation As String
    Dim strStartDate As String, strEndDate As String, strBody As String
    Dim XMLDoc As DOMDocument, objXML1 As Object, objXMLChild As Object

    '由於 Windows CE 的 MSXML 並不支援 xml 中 encoding 的屬性
    '所以必須從回傳的 XML 字串中，將 encoding 屬性去除掉，再利用 XMLDOM 解析內容
    strXML = Mid(strXML, 1, InStr(1, strXML, "encoding=") - 1) & _
        Mid(strXML, InStr(1, strXML, "?>"), Len(strXML) - InStr(1, strXML, "?>") + 1)
    Set XMLDoc = CreateObject("Microsoft.XMLDOM")
    XMLDoc.loadXML (strXML)

    Set objXML1 = XMLDoc.getElementsByTagName("Appointment")
    '解析出活動內容並依序新增至 TreeView 控制項(tvcItems)中
    For nTemp = 1 To objXML1.length
        Set objXMLChild = objXML1.Item(nTemp - 1)
        strSubject = objXMLChild.selectSingleNode("Subject").Text
        strBody = "內容：" & objXMLChild.selectSingleNode("Body").Text
        strStartDate = "日期：" & objXMLChild.selectSingleNode("Start").Text
        strEndDate = objXMLChild.selectSingleNode("End").Text
        strLocation = "地點：" & objXMLChild.selectSingleNode("Location").Text
        strProvider = "提供者：" & objXMLChild.selectSingleNode("Provider").Text
        strTemp = CStr(nCurrentProviderIndex) & CStr(nTemp)
        '新增 TreeView 的節點，在後面的文章會說明 TreeView 的語法
        tvcItems.Nodes.Add "Root", 1, "Root" & strTemp, strSubject
        tvcItems.Nodes.Add "Root" & strTemp, 4, "Date" & strTemp, _
            strStartDate & "~" & strEndDate
        tvcItems.Nodes.Add "Root" & strTemp, 4, "Body" & strTemp, strBody
        tvcItems.Nodes.Add "Root" & strTemp, 4, "Provider" & strTemp, strProvider
        tvcItems.Nodes.Add "Root" & strTemp, 4, "Location" & strTemp, strLocation
    Next
    ParseAppItems = True
End Function

```

程式碼 2 · ParseAppItems 函式

管理行事曆

利用 Pocket Outlook Object Model(POOM) 來管理行事曆(Calendar)、國家\地區(Cities)、連絡人(Contacts)、紅外線(Infrared)、工作(Tasks)。




圖 8 · Pocket Outlook 物件模型

加入 POOM 參考

您將不會在有效的參考(Available Reference)下找到[Microsoft Pocket Outlook Object Model v1.0] 程式庫，如果您有安裝 eVB3.0 的話，您可以在您的\Windows CE Tools\wce300\MS Pocket PC\emulation\palm300\windows\目錄下找到 pimstore.dll 檔案。

按下[Project]→[References] →[Browse...], 選擇\Windows CE Tools\wce300\MS Pocket PC\emulation\palm300\windows\pimstore.dll, 按下[開啓]。將會出現一個錯誤訊息，請按下[是](見圖 6)。您將可以把 Microsoft Pocket Outlook Object Model v1.0 程式庫參考進來，找到程式庫後，將名稱左方的核取方塊打勾，再按下[OK]鍵離開此畫面(見圖 9)。POOM SDK 下載處 <http://www.microsoft.com/mobile/downloads/developer/poomsdk.asp>。




圖 9 · 加入 POOM 參照

新增約會至行事曆

在使用 Application 物件任何的方法、屬性之前，都必須要先登入，語法如下：

```
objApplication.Logon
```

在最後，必須要登出，語法如下：

```
objApplication.Logoff
```

新增 Application 物件其子項目的語法：

```
objApplication.CreateItem(ItemType)
```

參數說明：

ItemType：新增項目的屬性，其屬性值如下：

1：約會， 2：連絡人， 3：工作， 101：國家/地區。

在本例中，需撰寫一個 AddAppointment 函式(見程式碼 3)，其功能為新增一項約會至行事曆並顯示這項約會。

```
Private Sub AddAppointment(ByVal strSubject As String, strBody As String, ByVal strLocation As String, ByVal strStartDate As String, strEndDate As String)
    Dim pol As PocketOutlook.Application
    Dim oAppointment As PocketOutlook.AppointmentItem
    Dim bSuccess As Boolean, nTemp As Integer, strTemp As String

    Set pol = CreateObject("PocketOutlook.Application")
    '登入 Pocket Outlook，將回傳成功失敗至 bSuccess 變數中
    bSuccess = pol.Logon
    '從 Application 物件中新增一個約會
    Set oAppointment = pol.CreateItem(1)
    '設定約會的主旨
    oAppointment.Subject = strSubject
    '設定約會的內容
    oAppointment.Body = strBody
    '設定約會的地點
    oAppointment.Location = strLocation
    '並將約會的狀態設為忙碌(Free=0, Tentative=1, Busy=2, Out of Office=3)
    oAppointment.BusyStatus = olBusy
    '若為整天的約會，則可設定 AllDayEvent 屬性為 true
    'oAppointment.AllDayEvent = True
    '設定約會開始時間
    oAppointment.Start = CDate(strStartDate)
    '設定約會結束時間
    oAppointment.End = CDate(strEndDate)
    '將這項約會儲存至行事曆中
    oAppointment.save

    If MsgBox("新增成功！是否要查看新增的活動？", vbYesNo, "新增成功") = vbYes Then
        '顯示這項新增的約會
    End If
End Sub
```

```

oAppointment.Display
End If
'登出
bSuccess = pol.Logoff
End Sub

```

程式碼 3 · 新增約會的函式

MenuBar , TreeView , PictureBox 控制項

請點選[Project]→[Components]，將 Microsoft CE PictureBox Control 3.0 、 Microsoft CE TreeView Control 3.0 、 Microsoft PocketPC MenuBar Control 3.0 這三項打勾，並按下[確定]，將這些 Active X 控制項參照進來(見圖 10)，在工具列就會多出三個 ActiveX 控制項，只要將控制項拖拉至表單上，就可以使用了。




圖 10 · 選擇 PictureBox 、 TreeView 、 PocketPc MenuBar 三個 ActiveX 控制項。

TreeView 控制項

TreeView 控制項可以顯示出樹狀結構，所以我們利用這個控制項來顯示活動的項目及內容，當使用者要將這項活動加入自己的行事曆時，只要選擇這項活動，並按下[新增至行事曆]選單，將會出現一個確認的訊息視窗，若選擇[確定]，這項活動將會加入您的行事曆當中。

TreeView 控制項新增一個 Node 物件的方法：

```
nodes.Add([Relative],[Relationship],[Key],Text,[Image],[SelectedImage])
```

參數說明：

Relative：選擇性。為父節點的名稱，若為根節點，則填"Root"。

Relationship：選擇性。為 relative 節點下的位置。Relationship 參數屬性值如下：

- tvwFirst=0：
最先的節點。此 Node 和在 relative 中所指名的節點位於同一層，並位於其它所有同層節點之前。
- tvwLast=1：
最後的節點。此 Node 和在 relative 中所指名的節點位於同一層，並位於其它所有同層節點之後。任何新加入的節點都會加在最後加入的節點之後。
- tvwNext= 2：
下一個節點，此為預設值。此 Node 位於在 relative 中所指名的節點之後。
- tvwPrevious=3：
前一個節點。此 Node 位於在 relative 中所指名的節點之前。
- tvwChild=4：
子節點。此 Node 成爲在 relative 中所指名節點的子節點。

Key：選擇性。爲這個節點的名稱，必須在這個節點集合裡是唯一的。

假如您新增一個 Key="Exit"的節點 `TreeView.Nodes.Add(, "Exit", "離開")`，你就可以用 `Treeview.Nodes.Item("Exit").屬性`，來取得該 Node 的引用。

Text：欲顯示於 Node 的文字，如本例中的“重新整理”、“重新連線”。

Image：選擇性。顯示於 Node 文字之前的圖示，這個參數爲另一個 ImageList 控制項的圖示編號。

SelectedImage：選擇性。當這個節點被選擇時，顯示於 Node 文字之前的圖示，這個參數爲另一個 ImageList 控制項的圖示編號。

當使用者下拉選單後(設此下拉選單的 Name='cmbProvider')，先改變 TreeView 的根節點，再呼叫 XML Web Service，去取得其詳細的活動項目，再新增子節點至 TreeView 中。

```
Private Sub cmbProvider_Click()  
    Dim strURL As String, strStartDate As String, strEndDate As String  
    Dim nCurrentProviderIndex As Integer  
    '取得被選擇的 Index  
    nCurrentProviderIndex = cmbProvider.SelectedIndex  
  
    If nCurrentProviderIndex >= 0 Then  
        tvcItems.Nodes.Clear  
        strStartDate = txtStartDate.Text  
        strEndDate = txtEndDate.Text  
        strCurrentProvider = strProviders(nCurrentProviderIndex, 2)  
        '新增 TreeView 控制項的根節點  
        tvcItems.Nodes.Add(, "Root", strProviders(nCurrentProviderIndex, 0))  
        strURL = strProviders(nCurrentProviderIndex, 1)  
        strURL = strURL & "/GetAppointments?strStartDate=" & strStartDate & _  
            "&strEndDate=" & strEndDate  
        Call GetWebService(strURL) '呼叫 XML Web Service(程式碼 1)  
    End If  
End Sub
```

程式碼 4 · 下拉式選單(cmbProvider)的 Click 事件

撰寫一個 AddCalendar 函式(程式碼 5)，方便之後其他程式使用，其程式功能為：當使用者選擇某一項活動後點選[新增至行事曆]時，必須先將 TreeView 中的活動內容、時間、地點、提供者取出，並傳至之前已經撰寫好的 AddAppointment 函式，新增約會至行事曆中。

```
Private Sub AddCalendar()  
    Dim Index As Long, nTemp As Integer, strTempValue As String  
    Dim strKey As String, strTempKey As String, strSubject As String, strBody As String  
    Dim strLocation As String, strProvider As String, strDate As String  
    Dim strStartDate As String, strEndDate As String  
    Dim objTemp As Node, objChd As Node  
  
    '取得 TreeView 中被選取的 Index  
    Index = tvclItems.SelectedItem.Index  
    '取其父節點  
    Set objTemp = tvclItems.Nodes.Item(Index).Parent  
    '若此節點沒有其父節點，則設定為選取的項目  
    If TypeName(objTemp) = "Nothing" Then  
        Set objTemp = tvclItems.Nodes.Item(Index)  
    End If  
  
    '取出節點的 Key  
    strKey = objTemp.Key  
    '取出節點的文字，即為活動的主旨  
    strSubject = objTemp.Text  
    If objTemp.Children > 0 Then  
        '陸續取出活動的內容、地點、提供者、活動期間  
        Set objChd = objTemp.Child  
        For nTemp = 1 To objTemp.Children  
            strTempKey = Replace(LCase(strKey), "root", "")  
            strTempKey = Replace(LCase(objChd.Key), strTempKey, "")  
            strTempValue = Mid(objChd.Text, InStr(1, objChd.Text, " : ") + 1, _  
                Len(objChd.Text) - InStr(1, objChd.Text, " : "))  
  
            Select Case strTempKey  
                Case "location"  
                    strLocation = strTempValue  
                Case "provider"  
                    strProvider = strTempValue  
                Case "body"  
                    strBody = strTempValue  
                Case "date"  
                    strDate = Split(strTempValue, "~")  
                    strStartDate = strDate(0)  
                    strEndDate = strDate(1)  
            End Select  
            Set objChd = objChd.Next  
        Next  
        If MsgBox("確定要將 [" & strSubject & "] 新增至您的行事曆中?", _  
            vbYesNo, "新增至行事曆") = vbYes Then  
            ' 呼叫自訂的函式，將新增約會至行事曆中(程式碼 3)  
            Call AddAppointment(strSubject, strBody, strLocation, strStartDate, strEndDate)  
        End If  
    End If  
End Sub
```

程式碼 5 · AddCalendar 函式

MenuBar 控制項

eVB 並沒有像 VB 一樣有視覺化的選單可以設定，必須利用 MenuBar 控制項來建立選單。而在本例中，必須建立一個選單，並在選單被點選時，做其選項的相對動作。

建立 MenuBar

我們要新增一個選單功能表(圖 11)，首先必須在 Form 上面先新增一個 MenuBar，在這裡我設定其 Name 為 myMenuBar。




圖 11 · 要建立的選單功能表

先用 myMenuBar 的 AddMenu 方法建立一個主 Menu，然後用主 Menu 的 Add 方法加入其他的子 Menu。

建立選單的語法：

```
set [變數名稱]=myMenuBar.Controls.AddMenu( [功能表標題],[功能表名稱])  
[變數名稱].Items.Add [不可重覆的索引值], [功能表名稱], [子功能表標題]
```

撰寫 MenuBar_Init 函式，建立選單(圖 11)：

```
Private Sub MenuBar_Init()  
    Dim myMenu As MenuBarItem  
  
    '新增一個 Menu，其顯示文字為"選項"  
    Set myMenu = myMenuBar.Controls.AddMenu("選項", "Option")  
    '在父 Menu 下，新增三個 MenuItem  
    myMenu.Items.Add 1, "Refresh", "重新整理"  
    myMenu.Items.Add 2, "ReConn", "重新連線"  
    myMenu.Items.Add 3, "Appointment", "新增至行事曆"  
    '建立功能表中的分隔線  
    myMenu.Items.Add 4, "Line", "-", mbrMenuItemSeparator  
    myMenu.Items.Add 5, "Exit", "離開"  
    Set myMenu = Nothing  
End Sub
```

程式碼 6 · MenuBar_Init 函式--新增 MenuBar

MenuBar 的 MenuItemClick 事件

當使用者點選選單時，會觸發 MenuItemClick 事件，在這個事件中，您可以自訂每個選項要做的動作。程式碼 7 中，將執行之前建立的選單中(程式碼 6)每個選項的動作。

```
Private Sub myMenuBar_MenuItemClick(ByVal Item As MenuItem) As Boolean  
    Select Case Item.Index
```

```

Case 1          '重新整理
    myMenuBar.Controls.Clear
    tvcItems.Nodes.Clear
    cmbProvider.Clear
    Form_Load
Case 2          '重新連線
    cmbProvider_Click
Case 3
    Call AddCalendar '新增至行事曆(程式碼 5)
Case Else
    App.End      '結束此程式
End Select
End Sub

```

程式碼 7 · 選擇 MenuBar 時，觸發 MenuClick 事件

PictureBox 控制項

這控制項主要是要顯示圖型或撥放*.wav 的聲音檔。在本例中，主要是利用 PictureBox 控制項來顯示圖形，只需要指定 Picture 的屬性即可，但是要注意的是，必須注意安裝之後的路徑，不然圖片找不到，這個程式可就完全沒有辦法執行了。

由於我們在安裝的時候，也要把這二個圖片拷貝至 PPC 上，所以必須將這二個圖片加入專的關聯檔，這樣在安裝的時候，安裝程式會把這些關聯檔完整的拷貝至 PPC 上。

請點選[File]→[Add File]，開始選擇圖片的位置，記得要把檔案類型改成 All Files(*.*)，並把視窗的左下方 Add As Related Document 的選項勾選起來(見圖 12)，加入之後，您便可以在專案視窗中看到此專案中所有的關聯檔(Related Document)(見圖 13)。




圖 12 · 加入圖片關聯檔




圖 13 · 在專案的視窗中，可看到加入的關聯檔。

由於安裝之後，我們並無法掌握圖片的路徑，所以必須先利用自己撰寫的 `AppPath` 函式(程式碼 7)來取出程式執行的路徑，再算出圖片的真實路徑。

```
Public Function AppPath()
    Dim path As String
    path = App.path
    '取得程式執行的路徑
    If path = "\" Then
        '如果是根目錄，就直接回傳\'
        AppPath = "\"
    Else
        AppPath = path & "\"
        '若不是，就回傳至本目錄並加上\'
    End If
End Function
```

程式碼 8 · `AppPath` 函式

撰寫 `Picture_Init` 函式，用於指定這二張圖片的路徑。

```
Private Sub Picture_Init()
    pboxProvider.Picture = AppPath & "Provider.bmp" 'AppPath 函式(程式碼 8)
    pboxCalendar.Picture = AppPath & "Calendar.bmp"
End Sub
```

程式碼 9 · `Picture_Init` 函式：設定圖片的路徑

Form_Onload 事件

在專案一開始會觸發 `Form_Onload` 事件，在這事件中，我們要將提供者的資料存至一個全域變數(`strProviders`)中，並將提供者名稱加入 `ComboBox` 下拉選單(`cmbProvider`)中。

```
Private Sub Form_Load()
    Dim strProvider As String, i As Integer, strTemp As Variant
    Call MenuBar_Init ' 建立選單 (程式碼 5)
    Call Picture_Init ' 指定圖片的路徑 (程式碼 8)

    '建立期間的初始值
    txtStartDate.Text = CStr(DatePart("yyyy", Date) & "/" & DatePart("m", Date) & _
        "/" & DatePart("d", Date))
```

```
strTemp = CDate(txtStartDate.Text) + 7
txtEndDate.Text = CStr(DatePart("yyyy", strTemp) & "/" & DatePart("m", strTemp) & _
    "/" & DatePart("d", strTemp))
```

建立提供者的下拉式選單

```
strProviders(0, 0) = "台灣微軟活動"
strProviders(0, 1) = "http://www.msdev.com.tw/MSTCWebService.asmx"
strProviders(1, 0) = "恆逸資訊微軟課程"
strProviders(1, 1) = "http://61.222.11.235/UCOMWebservice.asmx"
For i = 0 To UBound(strProviders, 1) - 1
    strProvider = strProviders(i, 0)
    If strProvider <> "" Then
        cmbProvider.AddItem strProviders(i, 0), i
    End If
Next
cmbProvider.ListIndex = 0
End Sub
```

程式碼 10 · Form_onload 事件

總結

透過這二期 XML Web Services 的實例整合範例，聰明的您是否也漸漸的了解到 XML Web Services 的魅力與強大的跨平台整合能力，相同的 XML Web Service 不僅能相容不同載具，也能整合在不同的平台中，這也是在下一個資訊技術里程碑中相當重要的發現。